

# GTAP6inGAMS: The Dataset and Static Model

Thomas F. Rutherford \*  
Ann Arbor, MI

Prepared for the Workshop:  
“Applied General Equilibrium Modeling for  
Trade Policy Analysis in Russia and the CIS”

The World Bank Resident Mission, Moscow

December 1-9, 2005

## 1 Introduction

The Global Trade Analysis Project (GTAP) is a research program initiated in 1992 to provide the economic research community with a global economic dataset for use in the quantitative analyses of international economic issues. The project’s objectives include the provision of a documented, publicly available, global, general equilibrium data base, and to conduct seminars on a regular basis to inform the research community about how to use the data in applied economic analysis.<sup>1</sup> GTAP has led to the establishment of a global network of researchers who share a common interest of multi-region trade analysis and related issues. The research program is coordinated by Thomas Hertel, Director of the Center for Global Trade Analysis at Purdue University. As Deputy Director of this Center, Robert McDougall oversees the data base work. Software development within the GTAP project has been assisted greatly by the efforts of Ken Pearson, Mark Horridge and other Australian researchers from Centre of Policy Studies, Monash University. (See Hertel, ed [1997] and McDougall [2005]). A list of applications based on the GTAP framework can be found at the GTAP home page, ([HTTP://WWW.GTAP.ORG](http://www.gtap.org)).

The principal programming language for GTAP data and modeling work is GEMPACK [Harrison and Pearson 1996]. In the GEMPACK framework the model is solved as a system of nonlinear equations. The present paper describes a third version of the GTAP model

---

\*This paper has been sponsored by the UK Department for International Development and the World Bank. The usual disclaimers apply.

<sup>1</sup>The GTAP version 6 database represents global production and trade for 89 country/regions, 59 commodities and 5 primary factors. The data characterize intermediate demand and bilateral trade in 2001, including tax rates on imports and exports and other indirect taxes.

implemented in GAMS. The GAMS model is essentially implemented as a nonlinear system of equations, although it can be posed either as a CNS (constrained nonlinear system) or MCP (mixed complementarity problem). Along with the core model I have developed two ancillary programs for dataset management. I call the package “GTAP6inGAMS”. These programs should be useful to economists who program in GAMS and wish to use GTAP in applied work. These programs include tools for translation of the GTAP files into GAMS readable form, GAMS programs for dataset aggregation and reconciliation. These tools permit users to easily filter and adjust tax rates on trade or domestic transactions.

This version of GTAPinGAMS differs in a couple of ways from Rutherford and Paltsev [2000] and Rutherford [1997]. First, this implementation more closely adheres to the GEMPACK implementation of GTAP 6. The GAMS model now accounts for differentiated import value shares in final demand, government demand and every intermediate demand coefficient. As in the GEMPACK implementation, the GAMS model now applies differentiated tax rates on all domestic and imported inputs. In addition, this implementation incorporates elasticities in production, trade, and factor supply which are consistent with the GEMPACK implementation.

There are a few substantive differences between the GEMPACK and GAMS version of the model. One of these is the final demand system. Whereas the GEMPACK model is based on a CDE demand system, the GAMS model employs Cobb-Douglas preferences. Second, there are differences in units of account. Values in the GAMS implementation differ from the GEMPACK model by a factor of 1000. The GTAP database measures all transactions in millions of dollars whereas GTAP6inGAMS measures transactions in billions of dollars. Third, the two models differ in their representation of investment demand and global capital markets. The GEMPACK model assumes that a “global bank” allocates international capital flows in response to changes in regional rates of return. The GTAP6inGAMS model makes the simplest possible assumptions regarding investment demand, international capital flows and the time path of adjustment: all of these variables are exogenously fixed at base year levels.

This paper consists of three sections following this overview. Section 2 introduces the dataset and core static model. I have written this material with the hope that that the model might be accessible to a second year graduate student who has never worked with GTAP. The exposition begins with a flow chart highlighting the source data. It then presents a “primal” version of the model formulation which emphasizes the role of optimizing behavior in the standard formulation.

Section 3 introduces the “dual” (Dixit-Norman) representation of the model’s equilibrium conditions, a formulation which closely resembles the model’s algebraic implementation in GAMS.

Section 4 presents the GAMS code itself, organized in a manner which highlights the close connection between the GAMS/MCP and GAMS/MPSGE versions of the model<sup>2</sup>. Complete documentation of the how the technology and preferences are calibrated is provided in the form of GAMS code which performs these tasks.

Section 5 has a practical perspective with step-by-step instructions on how to install the GTAP6inGAMS package. I hope this provides a short learning curve for economists who wish to perform a few calculations using the GTAP dataset. This section describes

---

<sup>2</sup>See Rutherford [1995] and Rutherford [1999] for introductions to these modeling formats

ancillary GAMS programs for dataset management.

## 2 The Model

The core GTAP model is a static, multi-regional model which tracks the production and distribution of goods in the global economy. In GTAP the world is divided into regions (often representing individual countries), and each region’s final demand structure is composed of public and private expenditure across goods. The model is based on optimizing behavior. Consumers maximize welfare subject to budget constraint with fixed levels of investment and public output. Producers combine intermediate inputs, and primary factors (skilled and unskilled labor, land, resources and physical capital) at least cost subject for given technology. The dataset includes a full set of bilateral trade flows with associated transport costs, export taxes and tariffs.

### 2.1 Benchmark Data and Accounting Identities

The economic structure underlying the GTAP model is illustrated in Figure 1. Symbols in this flow chart correspond to variables in the economic model.  $Y_{ir}$  portrays the production of good  $i$  in region  $r$ ,  $C_r$ ,  $I_r$  and  $G_r$  portray private consumption, investment and public demand, respectively.  $M_{jr}$  portrays the import of good  $j$  into region  $r$ .  $HH_r$  and  $GOVT_r$  stand for representative household and government consumers, and  $FT_{sr}$  is the activity through which “sluggish” factors of production are allocated to individual sectors.

In this figure commodity and factor market flows appear as solid lines. Domestic and imported goods markets are represented by horizontal lines at the top of the figure. Domestic production ( $vom_{ir}$ ) is distributed to exports ( $vamd_{irs}$ ), international transportation services ( $vst_{ir}$ ), intermediate demand ( $vdm_{ijr}$ ), household consumption ( $vdpm_{ir}$ ), investment ( $vdim_{ir}$ )<sup>3</sup> and government consumption ( $vdgm_{ir}$ ). The accounting identity in the GTAP6 dataset for domestic output is therefore:

$$vom_{ir} = \sum_s vamd_{irs} + vst_{ir} + \sum_j vdm_{ijr} + vdpm_{ir} + vdgm_{ir} + vdim_{ir}$$

Imported goods which have an aggregate value of  $vim_{ir}$  enter intermediate demand ( $vifm_{ijr}$ ), private consumption ( $vipm_{ir}$ ) and public consumption ( $vigm_{ir}$ ). The accounting identity for these flows is thus:

$$vim_{ir} = \sum_j vifm_{ijr} + vipm_{ir} + vigm_{ir}$$

Inputs to  $Y_{ir}$  include intermediate inputs (domestic and imported), mobile factors of production ( $vfm_{fir}$ ,  $f \in m$ ), and sluggish factors of production ( $vfr_{fir}$ ,  $f \in s$ ). Factor earnings accrue to households. Factor market equilibrium is given by an identity relating the value of factor payments to factor income:

$$\sum_i vfm_{fir} = evom_{fr}$$

---

<sup>3</sup> $vdim_{ir}$  is a symbol which has been introduced in the GAMS implementation of GTAP, primarily as a means of simplifying model exposition but also to provide greater flexibility in model aggregation.

International market clearance conditions require that region  $r$  exports of good  $i$  ( $vxm_{ir}$  at the top of the figure) equal the imports of the same good in all trading partners ( $vxml_{irs}$  at the bottom of the figure):

$$vxm_{ir} = \sum_s vxml_{irs}$$

Likewise, market clearance conditions apply for international transportation services. Aggregate provision of transportation service  $j$ ,  $vt_j$ , is equal to the value of transportation service exports:<sup>4</sup>

$$vt_j = \sum_r vst_{jr}$$

The supply-demand balance in the market for transportation service  $j$  equates transport services supply to the sum across all bilateral trade flows of service inputs ( $vtwr_{jisr}$  at the bottom of the figure):

$$vt_j = \sum_{isr} vtwr_{jisr}$$

Tax revenues and transfers in Figure 1 are indicated by dotted lines. The flows labelled with  $\mathcal{R}$  correspond to tax revenues.<sup>5</sup> Tax flows consist of indirect taxes on production/exports, ( $\mathcal{R}_{ir}^Y$ ), consumption ( $\mathcal{R}_r^C$ ), public demand ( $\mathcal{R}_r^G$ ) and imports ( $\mathcal{R}_{ir}^M$ ). Government income also includes a “direct tax” on households, portrayed by  $\mathcal{R}_r^{HH}$ , and net transfers from abroad,  $vb_r$ . The public budget constraint is thus given by:

$$\sum_i \mathcal{R}_{ir}^Y + \mathcal{R}_r^C + \mathcal{R}_r^G + \sum_i \mathcal{R}_{ir}^M + \mathcal{R}_r^{HH} + vb_r = vgm_r$$

Household budget constraint relates factor income net of tax payments to consumption expenditure and private investment:<sup>6</sup>

$$\sum_f evom_{fr} - \mathcal{R}_r^{HH} = vpm_r + vim_r$$


---

<sup>4</sup>Note that transportation service exports are treated as imperfect substitutes in the equilibrium model. This equation is an accounting identity which relates the value of services supply to the value of service “inputs”.

<sup>5</sup>These revenues do not appear as explicit variables in the GTAP database and are defined on the basis of expenditures and tax rates, as is subsequently described.

<sup>6</sup>For simplicity, international capital flows are portrayed as part of the public budget and any difference between household savings and investment is represented by implicit transfers in  $\mathcal{R}_r^{HH}$

Figure 1: Regional Economic Structure



To this point I have outlined two types of consistency conditions which are part of the GTAP database: market clearance (supply = demand for all goods and factors), and income balance (net income = net expenditure). A third set of identities involve net operating profits by all sectors in the economy. In the core GTAP model “production” takes place under conditions of perfect competition with constant returns to scale, hence there are no excess profits, and the cost of inputs must equal the value of outputs. This condition applies for each of the production sectors:

$$Y_{ir}: \sum_f vfm_{fir} + \sum_j (vifm_{jir} + vdfm_{jir}) + \mathcal{R}_{ir}^Y = vom_{ir}$$

$$M_{ir}: \sum_s (vxmd_{isr} + \sum_j vtwr_{jisr}) + \mathcal{R}_{ir}^M = vim_{ir}$$

$$C_r: \sum_i (vdpm_{ir} + vipm_{ir}) + \mathcal{R}_{ir}^C = vpm_r$$

$$G_r: \sum_i (vdgm_{ir} + vigm_{ir}) + \mathcal{R}_{ir}^G = vgm_r$$

$$I_r: \sum_i vdim_{ir} = vim_r$$

$$FT_{fr}: evom_{fr} = \sum_i vfm_{fir} \quad f \in s$$

$$YT_j: \sum_r vst_{jr} = vt_j = \sum_{irs} vtwr_{jirs}$$

## 2.2 The Primal Formulation

The notation used in the model is summarized in the Tables 1 - 3. Table 1 defines the various dimensions which characterize an instance of the model, including the set of sectors/commodities, the set of regions, the set of factors of production. The GTAP database includes a 89 regions and 59 commodities, but dimensionality typically limits the number of regions and goods which can be included in a single model.<sup>7</sup> Table 2 defines the primal variables (activity levels) which define an equilibrium. The model determines values of all the variables except international capital flows, a parameter which would be determined endogenously in an intertemporal model.

Table 3 defines the relative price variables for goods and factors in the model. As is the case in any Shoven-Whalley model, the equilibrium conditions determine *relative* rather than *nominal* prices. One market equilibrium condition corresponds to each of the equilibrium prices.

The benchmark identities presented in the previous section indicate the market clearance, zero profit and income balance conditions which define the GTAP model. The displayed equations do not, however, characterize the behaviour of agents in the model. In the competitive equilibrium setting, the standard assumption of optimizing atomistic agents applies for both producers and consumers. Profit maximization in the constant returns to scale setting is equivalent to cost minimization subject to technical constraints. For sector  $Y_{ir}$  we characterize input choices as though they arose from minimization of

---

<sup>7</sup>Datasets are easily aggregated which permits empirical applications to be formulated and debugging with small dimensional models.

Table 1: Set Indices

$i, j$	Sectors, an aggregation of the 55 sectors in the GTAP 5 database
$r$	Regions, an aggregation of the 49 regions in the GTAP 5 database
$f$	Factors of production (consisting of <i>mobile factors</i> , $f \in m$ , skilled labor, unskilled labor and capital, and <i>sluggish factors</i> , $f \in s$ , agricultural land and other resources)

Table 2: Activity Levels

$C_r$	Aggregate consumption demand
$G_r$	Aggregate public demand
$Y_{ir}$	Production.
$M_{ir}$	Aggregate imports
$FT_{fr}$	Factor transformation
$YT_j$	International transportation services

Table 3: Prices


$p_r^C$	Consumer price index
$p_r^G$	Public provision price index
$p_{ir}^Y$	Supply price, gross of indirect producer taxes
$p_{ir}^M$	Import price, gross of export taxes and tariffs.
$p_j^T$	Marginal cost of transport services
$p_{fr}^F$	Factor prices for labor, land and resources
$p_{fir}^S$	Price of sector-specific primary factors

Table 4: Tax and Subsidy Rates (net basis unless noted)

		Symbol	GAMS Parameter
Output taxes (gross basis)		$t_{ir}^o$	<code>rto(i,r)</code>
Factor taxes		$t_{fjr}^f$	<code>rtf(f,j,r)</code>
Intermediate input taxes	Domestic	$t_{ijr}^{fd}$	<code>rtfd(i,j,r)</code>
	Imported	$t_{ijr}^{fi}$	<code>rtfi(i,j,r)</code>
Consumption taxes	Domestic	$t_{ir}^{pd}$	<code>rtpd(i,r)</code>
	Imported	$t_{ir}^{pi}$	<code>rtpi(i,r)</code>
Public demand taxes	Domestic	$t_{ir}^{gd}$	<code>rtgd(i,r)</code>
	Imported	$t_{ir}^{gi}$	<code>rtgi(i,r)</code>
Export subsidies		$t_{isr}^{xs}$	<code>rtxs(i,s,r)</code>
Import tariffs		$t_{isr}^{ms}$	<code>rtms(i,s,r)</code>



Figure 2: Production Function :  $Y_{ir} = F_{ir}(ddf m, dif m, df m)$



unit costs:<sup>8</sup>

$$\min_{dif m, ddf m, df m} c_{ir}^D + c_{ir}^M + c_{ir}^F \quad (1)$$

s.t.

$$c_{ir}^D = \sum_j py_{jr}(1 + t_{jir}^{fd}) ddf m_{jir}$$

$$c_{ir}^M = \sum_j pm_{jr}(1 + t_{jir}^{fi}) dif m_{jir}$$

$$c_{ir}^F = \sum_f (pf_{fjr}|_{f \in m} + ps_{fjr}|_{f \in s})(1 + t_{fir}^f) df m_{fir}$$

$$F_{ir}(ddf m, dif m, df m) = Y_{ir}$$


The production function appearing in these constraints is described by a nested constant-elasticity-of-substitution (CES) form, as shown in Figure 2. The specific source of tax revenue is indicated in this figure, consisting of output taxes, taxes on intermediate inputs and taxes on factor demands, all of which are applied on a ad-valorem basis.

The choice among imports from different trading partners is based on Armington's idea of regionally differentiated products. The following cost minimization problem:

$$\min_{dxmd, dtwr} \sum_s (1 + t_{isr}^{ms}) \left( py_{is}(1 - t_{isr}^{xs}) dxmd_{isr} + \sum_j pt_j dtwr_{jisr} \right) \quad (2)$$

<sup>8</sup>Decision variables appearing in the primal model correspond to the benchmark data structures with the initial "v" replaced by "d". Hence, while  $vdf m_{jir}$  represents benchmark intermediate demand for good  $j$  in the production of good  $i$  in region  $r$ ,  $ddf m_{jir}$  represents the corresponding intermediate demand variable in the equilibrium producer's decision problem.

Figure 3: Armington Aggregation :  $M_{ir} = A_{ir}(dxmd, dtwr)$



s.t.

$$A_{ir}(dxmd, dtwr) = M_{ir}$$

The import aggregation function portrayed by  $A$  in (2) is described by the nested CES-Leontief function shown in Figure 3. Transportation services enter on a proportional basis with imports from different countries, reflecting differences in unit transportation margins across different goods and trading partners. Substitution at the top level in an Armington composite trades off imported goods (inclusive of associated transportation services). Trade flows are subject to export subsidies and import tariffs, with subsidies paid by government in the exporting region, and tariffs collected by government in the importing region.

Private consumption consistent with utility maximization is portrayed by minimization of the cost of a given level of aggregate consumption:

$$\min_{ddpm, dipm} \sum_i py_{ir}(1 + t_{ir}^{pd})ddpm_{ir} + pm_{ir}(1 + t_{ir}^{pi})dipm_{ir} \quad (3)$$

s.t.

$$H_r(ddpm, dipm) = C_{ir}$$

Final demand in the core model is characterized by a Cobb-Douglas tradeoff across composite goods which include both domestic and imported inputs. The nested CD-CES function is displayed in Figure 4.

Land and natural resources are portrayed as sector-specific factors of production supplied through constant-elasticity-of-transformation (CET) production function allocates composite factors to sectoral markets. The supply of sectoral factors of production are portrayed as arising from the following profit-maximization problem:

$$\max_{dfm} \sum_j dfm_{sjr} p_{s_jr} \quad (4)$$

Figure 4: Private Consumption :  $C_r = H_r(ddpm, dipm)$





Figure 5: Sluggish Factor Transformation :  $FT_f = \Gamma_f(dfm)$



s.t.

$$\Gamma_{sr}(dfm) = evom_{sr}$$

This CET function ( $\Gamma$ ) is illustrated in Figure 5.

International transportation services are provided as an aggregation of transportation services exported from countries throughout the world. The aggregation of transportation services is represented in the model by a cost minimization problem:

$$\min_{dst} \sum_r py_{ir} dst_{ir}$$

s.t.

$$T_i(dst) = YT_i$$

The aggregation function which combines transport service exports from multiple regions. The functional form which aggregates services from different regions is illustrated in Figure 6.

Figure 6: International Transportation Services :  $YT_j = T_j(dst)$





Figure 7: Public Consumption :  $G_r = G_r(ddgm, digm)$



Public consumption in the model is represented as a fixed coefficient (Leontief) aggregation of domestic-import composites. This formulation introduces substitution at the second level between domestic and imported inputs while holding sectoral commodity aggregates constant. Figure 7 illustrates the functional form.

### 3 Equilibrium Conditions

Equilibrium conditions in this version of the model are presented in a somewhat different format than in Rutherford [1997]. The formulation presented here is based on an explicit “dual” approach. (See, e.g., Dixit and Norman [1992].)

An Arrow-Debreu model concerns the interaction of consumers and producers in markets. Mathiesen [1985] proposed a representation of this class of models in which two types of equations define an equilibrium: zero profit and market clearance. The corresponding variables defining an equilibrium are activity levels (for constant-returns-to-scale firms) and commodity prices.<sup>9</sup> Here I augment Mathiesen’s framework with a third class of variables corresponding to consumer income levels. Commodity markets encompass primary endowments of households as well as producer outputs and imports. In equilibrium the aggregate supply of each good must be at least as great as total intermediate and final demand. Initial endowments are exogenous. Producer supplies and demands are defined by producer activity levels and relative prices. Final demands are determined by market prices.

Economists who have worked with conventional textbook equilibrium models can find Mathiesen’s framework to be somewhat opaque because many quantity variables are not explicitly specified in the model. Variables such as final demand by consumers, factor demands by producers and commodity supplies by producers, are defined implicitly in Mathiesen’s model. For example, given equilibrium prices for primary factors, consumer incomes can be computed, and given income and goods prices, consumers’ demands can

<sup>9</sup>Under a maintained assumption of perfect competition, Mathiesen may characterize technology as CRTS without loss of generality. Decreasing returns are accommodated through introduction of a specific factor, while increasing returns are inconsistent with the assumption of perfect competition. In this environment zero excess profit is consistent with free entry for atomistic firms producing an identical product.

then be determined. The consumer demand functions are written down in order to define an equilibrium, but quantities demanded need not appear in the model as separate variables. The same is true of inputs or outputs from the production process: relative prices determine conditional demand, and conditional demand times the activity level represents market demand. Omitting decision variables and suppressing definitional equations corresponding to intermediate and final demand provides significant computational advantages at the cost of a somewhat more complex model statement.<sup>10</sup>

### 3.1 Zero profit (arbitrage) conditions

All production activities in the model are represented by constant-returns-to-scale technologies, and markets are assumed to operate competitively with free entry and exit. As a consequence, equilibrium profits are driven to zero and the price of output reflects the cost of inputs. The following sets of equations relating output price to marginal cost are part of the definition of an equilibrium.<sup>11</sup>

The calculation of unit cost and unit revenue functions involves the definition of a number of ancillary variables. I follow the notational convention that variables which do not appear explicitly as GAMS variables are defined in un-numbered equations. This indicates that they are “optional” in the sense that they may be substituted out of the central system of equations which define the market clearance, zero profit and income balance equations. Price indices, unit cost and unit revenue indices, tax revenue equations, and compensated demand and supply functions are “invisible” in the MPSGE implementation of the model.

**N.B.** I use the symbol  $\theta$  to portray value shares from the base year data. In many cases subscripts on these value shares are omitted in order to economize on notation.

#### 3.1.1 Sectoral production ( $y_{jr}$ )

The unit cost of value-added is a CES composite of skilled and unskilled labor, land, resources and capital inputs to production, gross of taxes. Factor inputs may be sector-specific (sluggish) or flexible:

$$p_{fjr}^{pf} = \begin{cases} p_{fr}^F \frac{(1+t_{fjr}^f)}{1+t_{fjr}^f} & f \in m \\ p_{fjr}^S \frac{(1+t_{fjr}^f)}{1+t_{fjr}^f} & f \in s \end{cases}$$

and the unit cost function is given by:

$$c_{jr}^f = \left( \sum_f \theta_f (p_{fjr}^{pf})^{1-\sigma} \right)^{1/(1-\sigma)}$$

<sup>10</sup>The algebraic implementation of the GTAP model are provided here in GAMS/MCP and GAMS/CNS. These formats can be used interchangeably for GTAP models, hence the PATH or CONOPT solvers would both be suitable solvers. The MCP solver MILES might also be used but may be incapable of processing large scale instances. The model uses GAMS-F, a function preprocessor for GAMS has been used to avoid the introduction of intermediate variables representing unit cost and demand functions.

<sup>11</sup>This model is essentially a nonlinear system of equations which corresponds to a special form of complementarity. We retain consistency with the complementarity format by expressing arbitrage conditions as “oriented equations” with marginal cost on the LHS and marginal revenue on the RHS.

The user cost of intermediate inputs, differs from the market price, due to the presence of taxes on intermediate inputs, but we define units so that both the consumer and producer prices of these purchases are unity at benchmark tax rates:

$$p_{ijr}^d = p_{ir}^Y \frac{1 + t_{ijr}^d}{1 + \bar{t}_{ijr}^d}$$

$$p_{ijr}^i = p_{ir}^M \frac{1 + t_{ijr}^i}{1 + \bar{t}_{ijr}^i}$$

A CES cost function then describes the minimum cost of a bundle of domestic and imported inputs to production, based on benchmark value shares and an elasticity of substitution  $\sigma = esubd_i$ :

$$c_{ijr}^i = \left( \theta_d (p_{ijr}^d)^{1-\sigma} + (1 - \theta_d) (p_{ijr}^i)^{1-\sigma} \right)^{1/(1-\sigma)}$$

Unit cost is then a Leontief (linear) composite of the costs of intermediate and primary factor inputs, based on base year value shares:

$$c_{jr}^y = \sum_i \theta_i c_{ijr}^i + \theta_f c_{jr}^f$$

Having formulated the unit cost function, it is possible to compactly portray the zero profit condition for  $y_{jr}$ . In equilibrium, the marginal cost of supply equals the market price, net of taxes:

$$cy_{jr} = py_{jr} \frac{1 - t_{jr}^o}{1 - \bar{t}_{jr}^o} \quad (5)$$

Compensated demand functions related to  $y_{ir}$  include those which define domestic and imported intermediates.<sup>12</sup>

$$ddf m_{ijr} = y_{jr} vdf m_{ijr} \left( \frac{c_{ijr}}{p_{ijr}^d} \right)^\sigma$$

and

$$dif m_{ijr} = y_{jr} vif m_{ijr} \left( \frac{c_{ijr}}{p_{ijr}^i} \right)^\sigma$$

We also will subsequently need to define demands for primary factors in sector  $y_{ir}$ :

$$df m_{fjr} = y_{jr} vfm_{fjr} \left( \frac{c_{fjr}}{p_{fjr}^f} \right)^\sigma$$

---

<sup>12</sup>These demand functions are needed in formulating market clearance conditions below.

### 3.1.2 International transportation services ( $yt_j$ )

The unit cost of a transportation service depends on the benchmark value shares of region-specific services through a Cobb-Douglas cost function. Under perfect competition with free entry, the unit cost of international transport services equals the equilibrium market price:

$$\prod_r (py_{jr})^{\theta_j} = pt_j \quad (6)$$

The export demand for transportation service  $j$  from region  $r$  may then be written as a closed-form function of relative prices and the aggregate provision of those services ( $yt_j$ ):

$$dst_{jr} = yt_j vst_{jr} \frac{pt_j}{py_{jr}}$$

### 3.1.3 Private demand ( $c_r$ )

In the core model the consumer price index represents a Cobb-Douglas composite of the domestic and imported goods prices gross of tax. The price indices for domestic and imported goods are given by:

$$p_{ir}^{dc} = p_{ir}^y \frac{1 + t_{ir}^{pd}}{1 + \bar{t}_{ir}^{pd}}$$

and

$$p_{ir}^{ic} = p_{ir}^m \frac{1 + t_{ir}^{pi}}{1 + \bar{t}_{ir}^{pi}}$$

The unit cost of good  $i$  is a CES cost function defined over these price indices, based on benchmark value shares and an elasticity of substitution  $\sigma = esubd(i)$ :

$$p_{ir}^c = \left( \theta (p_{ir}^{dc})^{1-\sigma} + (1 - \theta) (p_{ir}^{ic})^{1-\sigma} \right)^{1/(1-\sigma)}$$

The consumer price of aggregate supply is defined by a Cobb-Douglas price index defined over price indices of domestic and imported varieties:

$$\prod_i (p_{ir}^c)^{\theta_i} = pc_r \quad (7)$$

Consumer demands for domestic and imported goods can be expressed on the basis of the aggregate consumption level and the prices of domestic and imported goods, gross of tax:

$$ddpm_{ir} = c_r v d p m_{ir} \left( \frac{p_{ir}^c}{p_{ir}^{dc}} \right)^\sigma \frac{pc_r}{p_{ir}^c}, \text{ and}$$

$$dipm_{ir} = c_r v i p m_{ir} \left( \frac{p_{ir}^c}{p_{ir}^{ic}} \right)^\sigma \frac{pc_r}{p_{ir}^c}.$$

### 3.1.4 Government demand ( $g_r$ ):

Public expenditure in the core model is a fixed-coefficient aggregate of Armington composite goods. Within each composite domestic and imported goods trade off with a constant elasticity of substitution. The unit price indices for domestic and imported goods are given by:

$$p_{ir}^{dg} = p_{ir}^y \frac{1 + t_{ir}^{dg}}{1 + \bar{t}_{ir}^{dg}}$$

and

$$p_{ir}^{ig} = p_{ir}^m \frac{1 + t_{ir}^{ig}}{1 + \bar{t}_{ir}^{ig}}$$

The composite price of the  $i$ th good is then:

$$p_{ir}^g = \left( \theta (p_{ir}^{dg})^{1-\sigma} + (1-\theta) (p_{ir}^{ig})^{1-\sigma} \right)^{1/(1-\sigma)}$$

The cost of public services ( $G_r$ ) is defined by the Leontief cost coefficients:

$$\sum_i \theta_i p_{ir}^g = pg_r \quad (8)$$

Government demands for domestic and imported goods is written as:

$$ddgm_{ir} = g_r v dgm_{ir} \left( \frac{p_{ir}^g}{p_{ir}^{dg}} \right)^\sigma \frac{pg_r}{p_{ir}^g}$$

and

$$digm_{ir} = g_r v igm_{ir} \left( \frac{p_{ir}^g}{p_{ir}^{ig}} \right)^\sigma \frac{pg_r}{p_{ir}^g}$$

### 3.1.5 Aggregate imports ( $m_{ir}$ )

The import cost index applies export taxes, trade and transport margins and import tariffs to the producer supply prices in exporting regions:

$$py_{isr}^m = p_{is}^y \frac{(1 - t_{isr}^{xs})(1 + t_{isr}^{ms})}{(1 - \bar{t}_{isr}^{xs})(1 + \bar{t}_{isr}^{ms})}$$

and the unit price of transportation services is given by:

$$pt_{jisr}^m = p_j^T \frac{1 + t_{isr}^{ms}}{1 + \bar{t}_{isr}^{ms}}$$

Transportation margins enter as fixed coefficients with bilateral trade flows, so the unit delivered price is a convex combination of the unit prices with weights corresponding to base year value shares:

$$pyt_{isr}^m = \theta py_{isr}^m + \sum_j \theta_j^T pt_{jisr}^m$$



Having formed a price index for bilateral imports from region  $s$ , the CES cost index can be defined on the basis of value shares and the elasticity of substitution across imports from different regions,  $\sigma = esubm(i)$ :

$$cim_{ir} = \left( \sum_s \theta_s (pyt_{isr}^m)^{1-\sigma} \right)^{1/(1-\sigma)}$$

The import activity ( $m_{ir}$ ) has a zero profit condition which relates the unit cost of imports to the market price of the import aggregate:

$$cim_{ir} = pm_{ir} \quad (9)$$

Import demands can then be expressed in closed form on the basis of the sectoral import level and the tax-inclusive prices of imports and transportation services:

$$dxmd_{isr} = m_{ir} vxmd_{isr} \left( \frac{pm_{ir}}{pyt_{isr}^m} \right)^\sigma$$

and

$$dtwr_{jisr} = m_{ir} vtwr_{jisr} \left( \frac{pm_{ir}}{pyt_{isr}^m} \right)^\sigma$$

### 3.1.6 Sluggish Factor Transformation ( $ft_{fr}$ )

The unit value of sector-specific factors is defined as a CET revenue function based on the base year value shares ( $\theta_j$ )

$$pvfm_{fr} = \left( \sum_j \theta_j ps_{fjr}^{1+\eta} \right)^{1/(1+\eta)} \quad f \in s$$

This defines the profit-maximizing allocation of factors to individual sectors. In equilibrium, the unit value of the aggregate factor is equal to the maximum unit earnings:

$$pf_{fr} = pvfm_{fjr} \quad f \in s \quad (10)$$

## 3.2 Market clearance

Supply-demand conditions apply to all goods and factors. Benchmark demand and supply quantities appear as scale factors in many of these equations, typically multiplied by activity levels which are equal to unity in the reference equilibrium.<sup>13</sup>

<sup>13</sup>While not crucial for representation of the model as a nonlinear system of equations, I follow the MCP convention in writing out the market clearance conditions. The equations are “oriented”, with supply variables on the LHS and demands on the RHS. Hence, the sense of the equation is *supply*  $\geq$  *demand*. In the core model equilibrium prices should always be positive, but in extensions of the standard model it might be quite common to introduce inequalities and complementary slackness, in which case the proper orientation of the equations is essential. Hence, in equilibrium should the price of a good be zero, economic equilibrium is then consistent with an associated market in which *supply*  $>$  *demand*.

### 3.2.1 Private consumption ( $pc_r$ )

Consumer demand in region  $r$  in the reference equilibrium is  $vpm(\mathbf{r})$ :

$$c_r vpm_r = \frac{HH_r}{pc_r} \quad (11)$$

### 3.2.2 Public consumption ( $pg_r$ )

Public demand in region  $r$  in the reference equilibrium is  $vgm(\mathbf{r})$ :

$$g_r vgm_r = \frac{GOVT_r}{pg_r} \quad (12)$$

### 3.2.3 Firm output ( $py_{ir}$ )

Aggregate output of good  $i$  in region  $r$  in the reference equilibrium is  $vom(\mathbf{i}, \mathbf{r})$ :

$$y_{ir} vom_{ir} = \sum_j ddfm_{ijr} + ddpm_{ir} + vdim_{ir} + ddgm_{ir} + \sum_s dxmd_{irs} + dst_{ir} \quad (13)$$

Investment demand for good  $i$  in region  $r$  is exogenously fixed at the benchmark level,  $vdim_{ir}$ .

### 3.2.4 Composite imports ( $pm_{ir}$ )

The aggregate value of imports of good  $i$  in region  $r$  in the reference equilibrium is  $vim(\mathbf{i}, \mathbf{r})$ :

$$m_{ir} vim_{ir} = \sum_j difm_{ijr} + dipm_{ir} + digm_{ir} \quad (14)$$

### 3.2.5 Transport services ( $pt_j$ )

The aggregate demand (and supply) for transport service  $j$  in the benchmark equilibrium is  $vtw(\mathbf{j})$ :

$$yt_j vtw_j = \sum_{isr} dtwr_{jisr} \quad (15)$$

### 3.2.6 Primary factors ( $pf_{fr}$ )

The aggregate demand (and supply) of primary factor  $f$  in region  $r$  is  $evom(\mathbf{f}, \mathbf{r})$ :

$$evom_{fr} = \begin{cases} \sum_j dfm_{fjr} & f \in m \\ evom_{fr} ft_{fr} & f \in s \end{cases} \quad (16)$$

### 3.2.7 Specific factors ( $ps_{fjr}$ )

The net value of benchmark payments to factor  $f$  in sector  $j$  in region  $r$  is  $vfm(\mathbf{f}, \mathbf{j}, \mathbf{r})$ :

$$vfm_{fjr} \left( \frac{ps_{fjr}}{pf_{fr}} \right)^\eta = dfm_{fjr} \quad (17)$$

### 3.3 Private Income ( $HH_r$ )

Benchmark household taxes net transfers are given by  $vta_x_r$ . Private household income is given by:<sup>14</sup>

$$HH_r = \sum_f pf_{fr} evom_{fr} - pc_n vta_x_r - \sum_i py_{ir} vdim_{ir} \quad (18)$$

The price  $pc_n$  appearing in this equation refers to the consumption level level in region  $r = n$ , the “numeraire region”, the units of which are used to denominate both transfers and international capital flows.

### 3.4 Public Budget ( $GOVT_r$ )

The base year current account deficit in region  $r$  is  $vb(\mathbf{x})$ , and region  $r = n$  corresponds to the “numeraire region” in which international capital flows are denominated:

$$GOVT_r = \mathcal{R}_r + pc_n (vb_r + vta_x_r + dta_x_r) \quad (19)$$

Tax revenue in region  $r$  consists of output taxes, intermediate demand taxes, factor taxes, final demand taxes, import tariffs and export subsidies:

$$\mathcal{R}_r = \mathcal{R}_r^o + \mathcal{R}_r^{fd} + \mathcal{R}_r^{fi} + \mathcal{R}_r^f + \mathcal{R}_r^{pd} + \mathcal{R}_r^{pi} + \mathcal{R}_r^{gd} + \mathcal{R}_r^{gi} - \mathcal{R}_r^{xs} + \mathcal{R}_r^{ms} \quad (20)$$

Each of these componets of tax revenue can be calculated as an ad-valorem or proportional tax rate times a market price times the quantity demanded or produced.

Taxes related to  $y_{ir}$  include output taxes:

$$\mathcal{R}_r^o = \sum_j t_{jr}^o vom_{jr} py_{jr} y_{jr},$$

tax revenue from intermediate inputs:

$$\mathcal{R}_r^{fd} = \sum_{ij} t_{ijr}^{fd} py_{ir} ddf m_{ijr}, \text{ and}$$

$$\mathcal{R}_r^{fi} = \sum_{ij} t_{ijr}^{fi} pm_{ir} dif m_{ijr},$$

and factor tax revenue:

$$\mathcal{R}_r^f = \sum_{fj} t_{fjr}^f pf_{fr} df m_{fjr}.$$

Taxes on household consumption of domestic and imported goods are:

$$\mathcal{R}_r^{pd} = \sum_i t_{ir}^{pd} py_{ir} ddp m_{ir},$$

and

$$\mathcal{R}_r^{pi} = \sum_i t_{ir}^{pi} pm_{ir} dip m_{ir}.$$

---

<sup>14</sup>In the model we endogenized  $vta_x_r$  when performing equal yield lumpsum transfers to compensate for changes in public sector tax revenue.

Taxes on public demand for domestic and imported goods are:

$$\mathcal{R}_r^{gd} = \sum_i t_{ir}^{gd} p y_{ir} d d g m_{ir}$$

and

$$\mathcal{R}_r^{gi} = \sum_i t_{ir}^{gi} p m_{ir} d i g m_{ir}.$$

Export subsidies (paid by the government in the exporting region) are:

$$\mathcal{R}_r^{xs} = \sum_{is} t_{irs}^{xs} p y_{ir} d x m d_{irs}$$

and import tariff revenues are given by:

$$\mathcal{R}_r^{ms} = \sum_{is} t_{isr}^{ms} \left( p y_{is} (1 - t_{isr}^{xs}) d x m d_{isr} + \sum_j p t_j d t w r_{jisr} \right)$$

## 4 The GAMS Code

This section presents the GAMS code with some commentarity which relates the algebraic GAMS expressions to the corresponding pieces of the GAMS/MPSGE tabular input.

### 4.1 Declarations

```
$title gtap6ingams -- an algebraic implementation

*      Invoke the function preprocessor:

$sysinclude gams-f

*      Define a default dataset:

$if not set ds $set ds gtapiea

*      Read the dataset using the utility program:

$batinclude gtap6data

*      We need to have a third symbol referencing commodities,
*      in addition to i and j:

alias (j,jj);

*      Following Michael Saunders suggestion, variables are
*      in upper case, all else lower.

positive variables
    C(r)          Consumption
    G(r)          Government demand
    Y(i,r)        Supply
    M(i,r)        Imports
    YT(j)         Transportation services
    FT(f,r)       Specific factor transformation

    PC(r)         Private consumption price index
    PG(r)         Public consumption price index
    PY(j,r)       Domestic output price
    PM(j,r)       Import price
    PT(j)         Transportation services
    PF(f,r)       Primary factors
    PS(f,j,r)     Sector-specific primary factors

    HH(r)         Representative private household
    GOVT(r)       Representative government;

equations
    prf_c(r)      Consumption
    prf_g(r)      Government demand
    prf_y(i,r)    Supply
    prf_m(i,r)    Imports
    prf_yt(j)     Transportation services
    prf_ft(f,r)   Factor transformation
```

mkt_pc(r)	Private consumption price index
mkt_pg(r)	Public consumption price index
mkt_py(j,r)	Domestic output price
mkt_pm(j,r)	Import price
mkt_pt(j)	Transportation services
mkt_pf(f,r)	Primary factors
mkt_ps(f,j,r)	Specific factor
inc_hh(r)	Representative household
inc_govt(r)	Public budget;

## 4.2 Zero profit (arbitrage) conditions

### 4.2.1 Sectoral production – $Y(j,r)$

```

parameter      thetaf(f,j,r)  Factor share of value added,
                thetad(i,j,r)  Domestic share of intermediate input,
                thetai(i,j,r)  Import share of intermediate input,
                theta_f(j,r)   Value added share of sectoral output;

alias (f,ff);
thetaf(f,j,r)$sum(ff,vfm(ff,j,r)*(1+rtf0(ff,j,r)))
    = vfm(f,j,r)*(1+rtf0(f,j,r)) / sum(ff,vfm(ff,j,r)*(1+rtf0(ff,j,r)));
thetad(i,j,r)$ (vdfm(i,j,r)*(1+rtfd0(i,j,r)) + vifm(i,j,r)*(1+rtfi0(i,j,r)))
    = vdfm(i,j,r)*(1+rtfd0(i,j,r)) /
      (vdfm(i,j,r)*(1+rtfd0(i,j,r)) + vifm(i,j,r)*(1+rtfi0(i,j,r)));

thetai(i,j,r)$vom(j,r)
    = (vdfm(i,j,r)*(1+rtfd0(i,j,r)) + vifm(i,j,r)*(1+rtfi0(i,j,r))) / vom(j,r);

theta_f(j,r)$vom(j,r) = sum(ff,vfm(ff,j,r)*(1+rtf0(ff,j,r))) / vom(j,r);

*      User cost indices for factors, domestic and imported
*      intermediate inputs:

p_pf(f,j,r) == (PF(f,r)$mf(f)+PS(f,j,r)$sf(f))*(1+rtf(f,j,r))/(1+rtf0(f,j,r));
p_d(i,j,r) == PY(i,r)*(1+rtfd(i,j,r))/(1+rtfd0(i,j,r));
p_i(i,j,r) == PM(i,r)*(1+rtfi(i,j,r))/(1+rtfi0(i,j,r));

*      Compensated (CES) cost functions:

cf(j,r) == sum(f$thetaf(f,j,r), thetaf(f,j,r) * p_pf(f,j,r)**(1-esubva(j))**(1/(1-esubva(j))));
ci(i,j,r) == ( (thetad(i,j,r) * p_d(i,j,r)**(1-esubd(i)) +
    (1-thetad(i,j,r)) * p_i(i,j,r)**(1-esubd(i)) )**(1/(1-esubd(i))) )$thetai(i,j,r);

*      Leontief cost function:

cy(j,r) == sum(i$thetai(i,j,r), thetai(i,j,r) * ci(i,j,r)) + (theta_f(j,r)*cf(j,r))$theta_f(j,r);

prf_y(j,r)$vom(j,r)..      cy(j,r) =e= PY(j,r) * (1-rto(j,r));

*      Demand functions:

```

```

ddfm(i,j,r) == (vdfm(i,j,r)*Y(j,r)*(ci(i,j,r)/p_d(i,j,r))**esubd(i))$vdfm(i,j,r);
difm(i,j,r) == (vifm(i,j,r)*Y(j,r)*(ci(i,j,r)/p_i(i,j,r))**esubd(i))$vifm(i,j,r);
dfm(f,j,r) == (vfm(f,j,r)*Y(j,r)*(cf(j,r)/p_pf(f,j,r))**esubva(j))$vfm(f,j,r);

```

\* Tax revenue:

```

revto(r) == sum(j$vom(j,r), rto(j,r) * vom(j,r) * PY(j,r) * Y(j,r));
revtfd(r) == sum((i,j)$vdfm(i,j,r), rtf(i,j,r)*PY(i,r)*ddfm(i,j,r));
revtfi(r) == sum((i,j)$vifm(i,j,r), rtfi(i,j,r)*PM(i,r)*difm(i,j,r));
revtf(r) == sum((f,j)$vfm(f,j,r), rtf(f,j,r)*PF(f,r)*dfm(f,j,r));

```

The corresponding statements in the MPSGE model are:

```

$prod:Y(j,r)$vom(j,r) s:0 i.tl:esubd(i) va:esubva(j)
o:PY(j,r) q:vom(j,r) a:GOVT(r) t:rto(j,r)
i:PY(i,r) q:vdfm(i,j,r) p:(1+rtd0(i,j,r)) i.tl: a:GOVT(r) t:rtfd(i,j,r)
i:PM(i,r) q:vifm(i,j,r) p:(1+rtfi0(i,j,r)) i.tl: a:GOVT(r) t:rtfi(i,j,r)
i:PS(sf,j,r) q:vfm(sf,j,r) p:(1+rtf0(sf,j,r)) va: a:govt(r) t:rtf(sf,j,r)
i:PF(mf,r) q:vfm(mf,j,r) p:(1+rtf0(mf,j,r)) va: a:govt(r) t:rtf(mf,j,r)

```

This is “production block” describes a two-level nested CES function in which the top-level elasticity of substitution is zero ( $s:0$ ), one nest is defined for each element of set  $i$ ,  $i.tl:esubd(i)$ , and a nest named “va:” is defined with an elasticity of substitution equal to  $esubva(j)$ . Taxes apply to all inputs, with tax revenues accruing to the  $GOVT(r)$  agent. Reference prices are defined on a gross of tax basis (see the  $P:$  fields), and reference quantities correspond to the net of tax value of commodity and factor demands.<sup>15</sup>

#### 4.2.2 International transportation services – $YT(j)$

\* Profit function for international transportation services:

```

prf_yt(j)$vtw(j)..
prod(r, PY(j,r)**(vst(j,r)/vtw(j))) =e= PT(j);

```

\* Demand function:

```

dst(j,r) == (vst(j,r)*YT(j)*PT(j)/PY(j,r))$vst(j,r);

```

The corresponding statements in the MPSGE model describe the provision of international transport:

```

$prod:YT(j)$vtw(j) s:1
o:PT(j) q:vtw(j)
i:PY(j,r) q:vst(j,r)

```

<sup>15</sup>For more information on the MPSGE modeling framework, see Rutherford [1999] or [www.mpsge.org](http://www.mpsge.org).

### 4.2.3 Private demand – C(r)

```

parameter      thetadpm(i,r)  Domestic share of private demand
                thetaipm(i,r)  Import share of private demand
                thetapm(i,r)   Value share of good in in private demand;

thetadpm(i,r)$ (vdpm(i,r)*(1+rtpd0(i,r))+vipm(i,r)*(1+rtpi0(i,r)))
               = vdpm(i,r)*(1+rtpd0(i,r)) /
                 (vdpm(i,r)*(1+rtpd0(i,r))+vipm(i,r)*(1+rtpi0(i,r)));
thetaipm(i,r)$ (vdpm(i,r)*(1+rtpd0(i,r))+vipm(i,r)*(1+rtpi0(i,r)))
               = vipm(i,r)*(1+rtpi0(i,r)) /
                 (vdpm(i,r)*(1+rtpd0(i,r))+vipm(i,r)*(1+rtpi0(i,r)));
thetapm(i,r) = (vdpm(i,r)*(1+rtpd0(i,r))+vipm(i,r)*(1+rtpi0(i,r))) / vpm(r);

*      User cost price indicies:

p_dc(i,r) == PY(i,r) * (1+rtpd(i,r)) / (1+rtpd0(i,r));
p_ic(i,r) == PM(i,r) * (1+rtpi(i,r)) / (1+rtpi0(i,r));

*      Unit cost functions (CES):

p_c(i,r) == (thetadpm(i,r)*p_dc(i,r)**(1-esubd(i)) +
             thetapm(i,r)*P_ic(i,r)**(1-esubd(i)))**1/(1-esubd(i));

prf_c(r)..   prod(i$thetapm(i,r), p_c(i,r)**thetapm(i,r)) =e= PC(r);

*      Demand functions:

ddpm(i,r)   == (vdpm(i,r)*C(r)*(p_c(i,r)/p_dc(i,r))**esubd(i)
               *(PC(r)/p_c(i,r)))$vdpm(i,r);
dipm(i,r)   == (vipm(i,r)*C(r)*(p_c(i,r)/p_ic(i,r))**esubd(i)
               *(PC(r)/p_c(i,r)))$vipm(i,r);

*      Tax revenue:

revtpd(r) == sum(i$vdpm(i,r), rtpd(i,r) * PY(i,r) * ddpm(i,r));
revtpi(r) == sum(i$vipm(i,r), rtpi(i,r) * PM(i,r) * dipm(i,r));

```

The corresponding statements in the MPSGE model describe the private consumption block:

```

$prod:C(r)  s:1  i.tl:esubd(i)
o:PC(r)    q:vpm(r)
i:PY(i,r)  q:vdpm(i,r)    i.tl: p:(1+rtpd0(i,r)) a:GOVT(r) t:rtpd(i,r)
i:PM(i,r)  q:vipm(i,r)    i.tl: p:(1+rtpi0(i,r)) a:GOVT(r) t:rtpi(i,r)

```

### 4.2.4 Government demand – G(r)

```

parameter      thetadgm(i,r)  Domestic value share in government demand
                thetaigm(i,r)  Import value share in government demand
                thetagm(i,r)   Aggregate value share in government demand;

thetadgm(i,r)$ (vdgm(i,r)*(1+rtdg0(i,r))+vigm(i,r)*(1+rtdgi0(i,r)))
               = vdgm(i,r)*(1+rtdg0(i,r)) /
                 (vdgm(i,r)*(1+rtdg0(i,r))+vigm(i,r)*(1+rtdgi0(i,r)));

```



```

thetaigm(i,r)$(vdgm(i,r)*(1+rtgd0(i,r))+vigm(i,r)*(1+rtgi0(i,r)))
      = vigm(i,r)*(1+rtgi0(i,r)) /
      (vdgm(i,r)*(1+rtgd0(i,r))+vigm(i,r)*(1+rtgi0(i,r)));
thetagm(i,r) = (vdgm(i,r)*(1+rtgd0(i,r))+vigm(i,r)*(1+rtgi0(i,r))) / vgm(r);

*      User cost price indices:

p_dg(i,r) == PY(i,r) * (1+rtgd(i,r)) / (1+rtgd0(i,r));
p_ig(i,r) == PM(i,r) * (1+rtgi(i,r)) / (1+rtgi0(i,r));

*      Unit cost functions (CES):

p_g(i,r) == (thetadgm(i,r)*p_dg(i,r)**(1-esubd(i)) +
      thetagm(i,r)*p_ig(i,r)**(1-esubd(i)))**1/(1-esubd(i));

*      Leontief unit cost function:

prf_g(r)..      sum(i, thetagm(i,r) * p_g(i,r) ) =e= PG(r);

*      Demand functions:

ddgm(i,r)  == (vdgm(i,r)*g(r)*(p_g(i,r)/p_dg(i,r))**esubd(i))$vdgm(i,r);
digm(i,r)  == (vigm(i,r)*g(r)*(p_g(i,r)/p_ig(i,r))**esubd(i))$vigm(i,r);

*      Tax revenue functions:

revtgd(r) == sum(i$vdgm(i,r), rtgd(i,r) * PY(i,r) * ddgm(i,r));
revtgi(r) == sum(i$vigm(i,r), rtgi(i,r) * PM(i,r) * digm(i,r));

```

The corresponding statements in the MPSGE model describe the public sector production block:

```

$prod:G(r)  s:0  i.tl:esubd(i)
o:PG(r)    q:vgm(r)
i:PY(i,r) q:vdgm(i,r) i.tl: p:(1+rtgd0(i,r)) a:GOVT(r) t:rtgd(i,r)
i:PM(i,r) q:vigm(i,r) i.tl: p:(1+rtgi0(i,r)) a:GOVT(r) t:rtgi(i,r)

```

#### 4.2.5 Aggregate imports – M(i,r)

```

*      User cost indices:

py_m(i,s,r) == PY(i,s) * (1-rtxs(i,s,r))*(1+rtms(i,s,r)) / pvxmd(i,s,r);
pt_m(j,i,s,r) == PT(j) * (1+rtms(i,s,r)) / pvtwr(i,s,r);

parameter      thetavxmd(i,s,r)      Value share of goods in imports,
                thetavtwr(j,i,s,r)    Value share of transportation services,
                thetam(i,s,r)         Bilateral import value share;

thetavxmd(i,s,r)$(vxmd(i,s,r)*pvxmd(i,s,r) + sum(j,vtwr(j,i,s,r)*pvtwr(i,s,r)))
      = vxmd(i,s,r)*pvxmd(i,s,r) /
      (vxmd(i,s,r)*pvxmd(i,s,r) + sum(j,vtwr(j,i,s,r)*pvtwr(i,s,r)));
thetavtwr(j,i,s,r)$(vxmd(i,s,r)*pvxmd(i,s,r) + sum(jj,vtwr(jj,i,s,r)*pvtwr(i,s,r)))
      = vtwr(j,i,s,r)*pvtwr(i,s,r) /
      (vxmd(i,s,r)*pvxmd(i,s,r) + sum(jj,vtwr(jj,i,s,r)*pvtwr(i,s,r)));
thetam(i,s,r)$vim(i,r)

```

```

= (vxmd(i,s,r)*pvxmd(i,s,r) + sum(j,vtwr(j,i,s,r)*pvtwr(i,s,r)))/vim(i,r);

* Price index of bilateral imports (Leontief cost function):

pyt_m(i,s,r) == py_m(i,s,r)*thetavxmd(i,s,r) + sum(j, pt_m(j,i,s,r)*thetavtwr(j,i,s,r));

* Unit cost function for imports (CES):

cim(i,r) == sum(s$thetam(i,s,r), thetam(i,s,r) * pyt_m(i,s,r)**(1-esubm(i))**(1/(1-esubm(i))));

prf_m(i,r)$vim(i,r).. cim(i,r) =e= PM(i,r);

* Demand functions:

dxmd(i,s,r) == (vxmd(i,s,r)*m(i,r)*(PM(i,r)/pyt_m(i,s,r))**esubm(i))$vxmd(i,s,r);
dtwr(j,i,s,r) == (vtwr(j,i,s,r)*m(i,r)*(PM(i,r)/pyt_m(i,s,r))**esubm(i))$vtwr(j,i,s,r);

* Regional tax revenue:

revtxs(r) == sum((i,s)$vxmd(i,r,s), -rtxs(i,r,s) * PY(i,r) * dxmd(i,r,s));
revtms(r) == sum((i,s)$thetam(i,s,r), rtms(i,s,r) *
(PY(i,s)*(1-rtxs(i,s,r))*dxmd(i,s,r) + sum(j, PT(j) * dtwr(j,i,s,r))));

```

The corresponding statements in the MPSGE model are those which describe the import aggregation activity:

```

$prod:M(i,r)$vim(i,r) s:esubm(i) s.tl:0
o:PM(i,r) q:vim(i,r)
i:PY(i,s) q:vxmd(i,s,r) p:pvxmd(i,s,r) s.tl:
+ a:GOVT(s) t:(-rtxs(i,s,r))
+ a:GOVT(r) t:(rtms(i,s,r)*(1-rtxs(i,s,r)))
i:PT(j)#(s) q:vtwr(j,i,s,r) p:pvtwr(i,s,r) s.tl:
+ a:GOVT(r) t:rtms(i,s,r)

```

#### 4.2.6 Sluggish Factor Transformation – FT(f,r)

```

parameter thetavfm(f,j,r) Value shares of specific factors;
thetavfm(sf,j,r) = vfm(sf,j,r)/evom(sf,r);

pvfm(sf,r) == sum(j, thetavfm(sf,j,r) * PS(sf,j,r)**(1+etrae(sf))**(1/(1+etrae(sf))));

prf_ft(sf,r)$evom(sf,r).. PF(sf,r) =e= pvfm(sf,r);

```

The corresponding statements in the MPSGE model involve the block which produces sector-specific factor inputs through a constant-elasticity-of-transformation function:

```

$prod:FT(sf,r)$evom(sf,r) t:etrae(sf)
o:PS(sf,j,r) q:vfm(sf,j,r)
i:PF(sf,r) q:evom(sf,r)

```

### 4.3 Market clearance

These equations do not appear explicitly in the MPSGE model, as they are generated automatically on the basis of the production function information provided above.

#### 4.3.1 Private consumption – PC(r)

mkt\_pc(r).. C(r) \* vpm(r) \* PC(r) =e= HH(r);

#### 4.3.2 Public consumption – PG(r)

mkt\_pg(r).. g(r) \* vgm(r) \* PG(r) =e= GOVT(r);

#### 4.3.3 Firm output – PY(i,r)

mkt\_py(i,r)\$vom(i,r)..  
y(i,r) \* vom(i,r) =e= sum(j, ddfm(i,j,r))  
+ ddpm(i,r) + ddgm(i,r)  
+ sum(s, dxmd(i,r,s)) + dst(i,r) + vdim(i,r);

#### 4.3.4 Composite imports – PM(i,r)

mkt\_pm(i,r)\$vim(i,r).. m(i,r) \* vim(i,r) =e= sum(j, difm(i,j,r)) + dipm(i,r) + digm(i,r);

#### 4.3.5 Transport services – PT(j)

mkt\_pt(j)\$vtw(j).. YT(j) \* vtw(j) =e= sum((i,s,r), dtwr(j,i,s,r));

#### 4.3.6 Primary factors – PF(f,r)

mkt\_pf(f,r)\$evom(f,r).. evom(f,r) =e= sum(j, dfm(f,j,r))\$mf(f) + (evom(f,r)\*FT(f,r))\$sf(f);

#### 4.3.7 Specific factors – PS(f,j,r)

mkt\_ps(sf,j,r)\$vfm(sf,j,r)..  
vfm(sf,j,r) \* (PS(sf,j,r)/PF(sf,r))\*\*etrae(sf) =e= dfm(sf,j,r);

#### 4.4 Household income balance – HH(r)

inc\_hh(r)\$HH.lo(r) <> HH.up(r)..  
HH(r) =e= sum(rnum, -pc(r)\*vtax(r))  
- sum(i, PY(i,r) \* vdim(i,r))  
+ sum(f, PF(f,r)\*evom(f,r));

The corresponding statements in the MPSGE model relate to the demand function for private households:

```
$demand:HH(r)
d:PC(r)          q:vpm(r)
e:PY(i,r)        q:(-vdim(i,r))
e:PF(f,r)        q:evom(f,r)
e:PC(r)          q:(-vtax(r))
```

## 4.5 Public budget – GOVT(r)

```
inc_govt(r)..
    GOVT(r) =e= sum(rnum, PC(rnum)*vb(r))+ PC(r)*vtax(r)
              + revto(r) + revtfd(r) + revtfdi(r) + revtf(r)
              + revtpd(r) + revtpti(r) + revtgd(r) + revtgi(r)
              + revtxs(r) + revtms(r);
```

Corresponding statements in the MPSGE model include the demand block for the public agent. (Note that tax revenues are automatically included in this agents budget constraint and do not appear in the demand function.):

```
$demand:GOVT(r)
d:PG(r)          q:(-vgm(r))
e:PC(rnum)       q:vb(r)
e:PC(r)          q:vtax(r)
```

## 4.6 Model Declaration and Benchmark Replication

```
model gtap6mcp /
    prf_c.c,prf_g.g,prf_y.y,prf_m.m,prf_yt.yt,prf_ft.ft,
    mkt_pc.pc,mkt_pg.pg,mkt_py.py,mkt_pm.pm,mkt_pt.pt,mkt_pf.pf,mkt_ps.ps,
    inc_hh.hh, inc_govt.govt/;

model gtap6cns /
    prf_c.c,prf_g.g,prf_y.y,prf_m.m,prf_yt.yt,prf_ft.ft,
    mkt_pc.pc,mkt_pg.pg,mkt_py.py,mkt_pm.pm,mkt_pt.pt,mkt_pf.pf,mkt_ps.ps,
    inc_hh.hh, inc_govt.govt/;

*      Assign default values:

c.l(r) = 1; g.l(r) = 1; y.l(i,r) = 1; m.l(i,r) = 1; yt.l(j) = 1;
ft.l(sf,r) = 1; pc.l(r) = 1; pg.l(r) = 1; py.l(j,r) = 1; pm.l(j,r) = 1;
pt.l(j) = 1; pf.l(f,r) = 1; ps.l(sf,j,r) = 1; HH.l(r) = vpm(r);
GOVT.L(r) = vgm(r);

*      Fix variables which should not enter the model:

y.fx(i,r)$(vom(i,r)=0) = 1; m.fx(i,r)$(vim(i,r)=0) = 1;
yt.fx(j)$(vtw(j)=0) = 1; py.fx(j,r)$(vom(j,r)=0) = 1;
pm.fx(j,r)$(vim(j,r)=0) = 1; pt.fx(j)$(vtw(j)=0) = 1;
pf.fx(f,r)$(evom(f,r)=0) = 1; ps.fx(f,j,r)$(not sf(f)) or (vfm(f,j,r)=0) = 1;
ft.fx(f,r)$(not sf(f)) or (evom(f,r)=0) = 1;

*      Establish a price normalization using the reference region:

HH.fx(rnum) = HH.l(rnum);

*      Verify benchmark consistency with both MCP and CNS models:

gtap6mcp.iterlim = 0;
solve gtap6mcp using mcp;

gtap6cns.iterlim = 0;
\solve gtap6cns using cns;
```

## 4.7 Declaration and Replication – MPSGE Version

```

$title Read GTAP6 Basedata and Replicate the Benchmark in MPSGE

*      Why not solve a model with the full GTAP6 dataset!
*      You don't even need a 64bit operating system, as
*      this model only requires 128MB of memory!

$set ds gsd01
$include gtap6data

$ontext
$model:gtap6

$sectors:
    c(r)                ! Consumption
    g(r)                ! Government demand
    y(i,r)$vom(i,r)     ! Supply
    m(i,r)$vim(i,r)     ! Imports
    yt(j)$vtw(j)        ! Transportation services
    ft(f,r)$(sf(f) and evom(f,r)) ! Specific factor transformation

$commodities:
    pc(r)                ! Private consumption price index
    pg(r)                ! Public consumption price index
    py(j,r)$vom(j,r)    ! Domestic output price
    pm(j,r)$vim(j,r)    ! Import price
    pt(j)$vtw(j)        ! Transportation services
    pf(f,r)$evom(f,r)  ! Primary factors rent
    ps(f,j,r)$(sf(f) and vfm(f,j,r)) ! Sector-specific primary factors

$consumers:
    ra(r)                ! Representative agent

$prod:y(j,r)$vom(j,r)  s:0  i.tl:esubd(i) va:esubva(j)
    o:py(j,r)           q:vom(j,r)   a:GOVT(r) t:rto(j,r)
    i:py(i,r)           q:vdfm(i,j,r) p:(1+rtrfd0(i,j,r)) i.tl: a:GOVT(r) t:rtrfd(i,j,r)
    i:pm(i,r)           q:vifm(i,j,r) p:(1+rtrfi0(i,j,r)) i.tl: a:GOVT(r) t:rtrfi(i,j,r)
    i:ps(sf,j,r)       q:vfm(sf,j,r) p:(1+rtrf0(sf,j,r)) va: a:GOVT(r) t:rtrf(sf,j,r)
    i:pf(mf,r)         q:vfm(mf,j,r) p:(1+rtrf0(mf,j,r)) va: a:GOVT(r) t:rtrf(mf,j,r)

$prod:yt(j)$vtw(j)  s:1
    o:pt(j)             q:vtw(j)
    i:py(j,r)           q:vst(j,r)

$prod:c(r)  s:1  i.tl:esubd(i)
    o:pc(r)           q:vpm(r)
    i:py(i,r)         q:vdpm(i,r)   i.tl: p:(1+rtrpd0(i,r)) a:GOVT(r) t:rtrpd(i,r)
    i:pm(i,r)         q:vipm(i,r)   i.tl: p:(1+rtrpi0(i,r)) a:GOVT(r) t:rtrpi(i,r)

$prod:g(r)  s:0  i.tl:esubd(i)
    o:pg(r)           q:vgm(r)
    i:py(i,r)         q:vdgm(i,r)   i.tl: p:(1+rtrgd0(i,r)) a:GOVT(r) t:rtrgd(i,r)
    i:pm(i,r)         q:vigm(i,r)   i.tl: p:(1+rtrgi0(i,r)) a:GOVT(r) t:rtrgi(i,r)

$prod:m(i,r)$vim(i,r)  s:esubm(i)  s.tl:0

```

```

o:pm(i,r)          q:vim(i,r)
i:py(i,s)          q:vxmd(i,s,r)  p:pvxmd(i,s,r) s.tl:
+                  a:GOVT(s) t:(-rtxs(i,s,r))
+                  a:GOVT(r) t:(rtms(i,s,r)*(1-rtxs(i,s,r)))
i:pt(j)#(s)        q:vtwr(j,i,s,r) p:pvtwr(i,s,r) s.tl:
+                  a:GOVT(r) t:rtms(i,s,r)

$prod:ft(sf,r)$evom(sf,r)  t:etrae(sf)
o:ps(sf,j,r)          q:vfm(sf,j,r)
i:pf(sf,r)           q:evom(sf,r)

$demand:HH(r)
d:PC(r)              q:vpm(r)
e:PY(i,r)            q:(-vdim(i,r))
e:PF(f,r)            q:evom(f,r)
e:PC(r)              q:(-vtax(r))

$demand:GOVT(r)
d:PG(r)              q:(-vgm(r))
e:PC(rnum)           q:(vb(r))
e:PC(r)              q:vtax(r)

$offtext
$sysinclude mpsgeset gtap6

gtap6.workspace = 128;
gtap6.iterlim = 0;
$include gtap6.gen
solve gtap6 using mcp;

```

## 4.8 Datasets

GTAPinGAMS datasets are stored in the GAMS Data eXchange (GDX) format. Data stored in this format may be freely transferred to header array format using Mark Horridge's `gdx2har.exe` program.<sup>16</sup> Data may also be transferred to Excel using any of several free utilities provided by GAMS Development Corporation (see, e.g., `gdxrw.exe`). Any GTAPinGAMS dataset may be aggregated into fewer regions, sectors and primary factors. This permits a modeller to do preliminary model development using a small dataset to ensure rapid response and a short debug cycle. After having implemented a small model, it is then a simple matter to expand the number of sectors and/or regions in order to obtain a more precise empirical estimate.

All GTAP datasets are defined in terms of three primary sets:  $i$ , the set of sectors and produced commodities,  $r$  the set of countries and regions, and  $f$  the set of primary factors. Table 3.4 presents the identifiers for the 59 GTAP 6 sectors in their most disaggregate form. These sectors may be aggregated freely to produce more compact dataset.<sup>17</sup>

Table 3.4 presents regional identifiers in the full dataset many of which correspond to standard UN three-character country codes. Users can define their own aggregations of

<sup>16</sup>See either

[HTTP://WWW.MONASH.EDU.AU/POLICY/GP-GAMS.HTM](http://www.monash.edu.au/policy/gp-gams.htm) or  
[HTTP://WWW.MPSGE.ORG/GDXHAR](http://www.mpsge.org/gdxhar).

<sup>17</sup>Unlike previous implementations, GTAP6inGAMS does *not* require a that investment appear as a distinct sector.

the GTAP data and use any labels to describe regions. For technical reasons, if a GTAP dataset is to be used with MPSGE, then regional identifiers can have at most 4 characters. Table 7 presents the three-character identifiers which I normally use for primary factors.

```

$title GTAP6DATA.GMS Read a GTAP 6 dataset

* Initial creation data: 2 August 2005 (tfr)

* Updates:

* 3 August 2005 - Introduced sets define sector-specific factors
* sf(f) and mobile factors mf(f) on the basis of elasticity
* etrae(f) which is read from default.prm.

$if not set ds $set ds china
$if not set datadir $set datadir "..\data\"
$setglobal datadir %datadir%
$if not set ghgdata $set ghgdata no
$setglobal ghgdata %ghgdata%

set i(*) Goods
    f(*) Factors
    r(*) Regions

$if exist %ds%.gdx $gdxin '%ds%.gdx'
$if not exist %ds%.gdx $gdxin '%datadir%ds%.gdx'
$load r f i

set
    rnum(r) Numeraire region,
    sf(f) Sluggish primary factors (sector-specific)
    mf(f) Mobile primary factors
    src Sources /domestic, imported/,
    ec Energy goods /
        ecoa Coal,
        eoil Crude oil,
        egas Natural gas,
        ep_c Refined oil products,
        eely Electricity,
        egdt Gas distribution /;

alias (r,s), (i,j);

parameters
    vdgm(i,r) Government - domestic purchases at market prices,
    vigm(i,r) Government - imports at market prices,
    vdpm(i,r) Private households - domestic purchases at market prices,
    vipm(i,r) Private households - imports at market prices
    vdim(i,r) Investment demand
    vfm(f,j,r) Endowments - Firms' purchases at market prices,
    vdfm(i,j,r) Intermediates - firms' domestic purchases at market prices,
    vifm(i,j,r) Intermediates - firms' imports at market prices,
    vxmd(i,r,s) Trade - bilateral exports at market prices,
    vst(i,r) Trade - exports for international transportation
    vtwr(i,j,r,s) Trade - Margins for international transportation at world prices;

```

```

$load vdgm vigm vdpm vipm vfm vdim
$load vdfm vifm vxmd vst vtwr

parameter
    evf(ec,i,r)      Volume of input purchases by firms (mtoe)
    evh(ec,r)        Volume of purchases by households (mtoe)
    evt(ec,r,r)      Volume of bilateral trade (mtoe)
    evq(ec,r)        Energy production (mtoe);

$load evf evh evt
evq(ec,r) = sum(i, evf(ec,i,r)) + evh(ec,r) + sum(s, evt(ec,r,s)-evt(ec,s,r));

set      srcs      Source /dom, imp/,
        ghg        Greenhouse gases /co2/;

parameter      eghg(ghg,i,srcs*,r)      Emissions of greenhouse gases (Gg CO2);
$if not %ghgdata%==no $load eghg
$if %ghgdata%==no eghg(ghg,i,srcs,j,r)=0; eghg(ghg,i,srcs,"hh",r)=0; eghg(ghg,i,srcs,"govt",r)=0;
display eghg;

parameter
    rto(i,r)          Output (or income) subsidy rates
    rtf(f,j,r)        Primary factor and commodity rates taxes
    rtpd(i,r)         Private domestic consumption taxes
    rtpi(i,r)         Private import consumption tax rates
    rtgd(i,r)         Government domestic rates
    rtgi(i,r)         Government import tax rates
    rtfd(i,j,r)       Firms domestic tax rates
    rtfi(i,j,r)       Firms' import tax rates
    rtxs(i,r,s)       Export subsidy rates
    rtms(i,r,s)       Import taxes rates;

$load rto rtf rtpd rtpi rtgd rtgi rtfd rtfi rtxs rtms

parameter
    esubd(i)          Elasticity of substitution (M versus D),
    esubva(j)         Elasticity of substitution between factors
    esubm(i)          Intra-import elasticity of substitution,
    etrae(f)          Elasticity of transformation,
    eta(i,r)          Income elasticity of demand,
    epsilon(i,r)      Own-price elasticity of demand;

$load esubd esubva esubm etrae eta epsilon

*      Declare some intermediate arrays which are required to
*      evaluate tax rates:

parameter      vdm(i,r)          Aggregate demand for domestic output,
                vom(i,r)          Total supply at market prices;

vdm(i,r) = vdpm(i,r) + vdgm(i,r) + sum(j, vdfm(i,j,r)) + vdim(i,r);
vom(i,r) = vdm(i,r) + sum(s, vxmd(i,r,s)) + vst(i,r);

parameter
    rtf0(f,j,r)      Primary factor and commodity rates taxes

```



```

        rtpd0(i,r)      Private domestic consumption taxes
        rtpi0(i,r)      Private import consumption tax rates
        rtgd0(i,r)      Government domestic rates
        rtgi0(i,r)      Government import tax rates
        rtf0(i,j,r)     Firms domestic tax rates
        rtfd0(i,j,r)   Firms' import tax rates
        rtfs0(i,r,s)   Export subsidy rates
        rtms0(i,r,s)   Import taxes rates;

rtpd0(f,j,r) = rtpd(i,r);
rtpi0(i,r) = rtpi(i,r);
rtgd0(i,r) = rtgd(i,r);
rtgi0(i,r) = rtgi(i,r);
rtfd0(i,j,r) = rtfd(i,j,r);
rtfi0(i,j,r) = rtfi(i,j,r);
rtxs0(i,r,s) = rtxs(i,r,s);
rtms0(i,r,s) = rtms(i,r,s);

parameter      vxmd(i,s,r)      Import price (power of benchmark tariff)
                pvtwr(i,s,r)    Import price for transport services;

vxmd(i,s,r) = (1+rtms0(i,s,r)) * (1-rtxs0(i,s,r));
pvtwr(i,s,r) = 1+rtms0(i,s,r);

parameter      bmprofit(i,r)    Zero profit check;

bmprofit(i,r) = round(vom(i,r)*(1-rto(i,r))
- sum(j, vdfm(j,i,r)*(1+rtfd(j,i,r))+vifm(j,i,r)*(1+rtfi(j,i,r)))
- sum(f, vfm(f,i,r)*(1+rtf(f,i,r))), 8);
display bmprofit;

parameter
    vtw(j)      Aggregate international transportation services,
    vpm(r)      Aggregate private demand,
    vgm(r)      Aggregate public demand,
    vim(i,r)    Aggregate imports,
    evom(f,r)   Aggregate factor endowment at market prices,
    vb(*)       Current account balance;

vtw(j) = sum(r, vst(j,r));
vpm(r) = sum(i, vdp(i,r)*(1+rtpd0(i,r)) + vipm(i,r)*(1+rtpi0(i,r)));
vgm(r) = sum(i, vdgm(i,r)*(1+rtgd0(i,r)) + vigm(i,r)*(1+rtgi0(i,r)));
vim(i,r) = vipm(i,r) + vigm(i,r) + sum(j, vifm(i,j,r));
evom(f,r) = sum(j, vfm(f,j,r));
vb(r) = vpm(r) + vgm(r) + sum(i, vdim(i,r))
- sum(f, evom(f,r))
- sum(j, vom(j,r)*rto(j,r))
- sum(j, sum(i, vdfm(i,j,r)*rtfd(i,j,r) + vifm(i,j,r)*rtfi(i,j,r)))
- sum(j, sum(f, vfm(f,j,r)*rtf(f,j,r)))
- sum(i, vdp(i,r)*rtpd(i,r) + vipm(i,r)*rtpi(i,r))
- sum(i, vdgm(i,r)*rtgd(i,r) + vigm(i,r)*rtgi(i,r))
- sum((i,s), rtms(i,s,r) * (vxmd(i,s,r) * (1-rtxs(i,s,r)) + sum(j,vtwr(j,i,s,r))))
+ sum((i,s), rtxs(i,r,s) * vxmd(i,r,s));

vb("chksum") = sum(r, vb(r));

```

```

display vb;

*      Determine which factors are sector-specific

mf(f) = yes$(1/etrae(f)=0);
sf(f) = yes$(1/etrae(f)>0);
display mf,sf;

parameter      mprofit Zero profit for m,
                yprofit Zero profit for y;

mprofit(i,r) = vim(i,r) - sum(s, pvxmd(i,s,r)*vxmd(i,s,r)+sum(j, vtwr(j,i,s,r))*pvtwr(i,s,r));
mprofit(i,r) = round(mprofit(i,r),5);
display mprofit;

yprofit(j,r) = vom(j,r)*(1-rto(j,r))-sum(i, vdfm(i,j,r)*(1+rtfd0(i,j,r))
        + vifm(i,j,r)*(1+rtfi0(i,j,r))) - sum(f, vfm(f,j,r)*(1+rtf0(f,j,r)));
yprofit(i,r) = round(yprofit(i,r),6)
display yprofit;

*      Determine which factors are sector-specific

mf(f) = yes$(1/etrae(f)=0);
sf(f) = yes$(1/etrae(f)>0);
display mf,sf;

parameter      mprofit Zero profit for m,
                yprofit Zero profit for y;

mprofit(i,r) = vim(i,r) - sum(s, pvxmd(i,s,r)*vxmd(i,s,r)+sum(j, vtwr(j,i,s,r))*pvtwr(i,s,r));
mprofit(i,r) = round(mprofit(i,r),5);
display mprofit;

yprofit(j,r) = vom(j,r)*(1-rto(j,r))-sum(i, vdfm(i,j,r)*(1+rtfd0(i,j,r))
        + vifm(i,j,r)*(1+rtfi0(i,j,r))) - sum(f, vfm(f,j,r)*(1+rtf0(f,j,r)));
yprofit(i,r) = round(yprofit(i,r),6)
display yprofit;

*      Define a numeraire region for denominating international
*      transfers:

rnum(r) = yes$(vpm(r)=smax(s,vpm(s)));
display rnum;

```

## 5 Practicalities

### 5.1 System Requirements

You will need to have the following:

- A computer.
- A recent GAMS system.

- The PATH or CONOPT solvers (installed GAMS subsystems).
- MPSGE subsystem (optional)

## 5.2 Getting Started

The GTAPinGAMS package is distributed as a zip file containing the directory structure and GAMS programs which unzips into a new directory.

GAMS source code for model management, several aggregate datasets, and some template applications are provided with the distribution directory. The GTAP source data is not distributed with the GTAPinGAMS system. In order to generate large scale models with the GTAPinGAMS tools, it is necessary to obtain the GTAP 6 distribution data files, either from the FlexAgg or GTAPAgg packages.<sup>18</sup> When working with FlexAgg, the following source data files are required:: GSDSET.HAR GSDDAT.HAR, GSDPAR.HAR and GSDVOLE.HAR. When working with GTAPAgg, any .ZIP archive can be translated into GTAP6inGAMS format.

Here are the steps involved in installing GTAP6inGAMS:

1. Create an empty root directory for the GTAP6 model.
2. Unzip GTAP6INGAMS.ZIP in the root directory.

The resulting directory should have the following subdirectories:

**data** This directory holds datasets in GTAPAgg (.ZIP), FlexAgg (.HAR) and GTAP6inGAMS (.GDX) formats. The distribution includes ASA7X5.GDX and CHP13.GDX.

**build** This directory contains tools for translating GTAP datasets to GDX format, filtering, aggregation and rebalancing.

**defines** Contains mapping files which define model aggregations. The syntax of these files may have changed from the previous (version 5) release. In the present model all datasets are self-contained GAMS data exchange (GDX) files. When aggregating a dataset, a single .MAP file defines the region, commodity and factor identifiers, and it also indicates how elements in the source model map to these items.

**model** This directory contains model source code and a few template applications.

3. Install the GTAP6 data and run the build script (optional)

The steps involved in building GTAP6 from scratch are documented in BUILD.BAT which is in the BUILD subdirectory.


4. Solve the sample models.

The steps involved in testing the installation are in the MODEL subdirectory. See RUN.BAT

---

<sup>18</sup>In addition to the economic data, the GTAP6inGAMS package can carry along carbon dioxide emissions data based on the recently released statistics:  
<https://www.gtap.agecon.purdue.edu/resources/download/2348.zip>

Figure 8: GTAP6 Model Dimensions



### 5.3 Aggregation and Dimensionality

Model development work should always be conducted with a small dataset. Only once the key simulations have been conducted and thoroughly analyzed in a small instance does it make sense to work with a larger number of regions or commodities.

There are computational limits to model size. Installed memory needs to be at least 1GB to deal with over 20 regions and 20 goods. It may be possible to work with larger models, but I've not experimented. My PC is capable of reading and performing a function evaluation with a the full (89 region / 59 sector) GTAP database, but I do not have enough memory to solve a model of this size.

Figure 8 provides a rough representation of how large a simultaneous system of nonlinear equations must be solved to deal with specified numbers of regions and commodities. (This graph is based on fairly rough estimates and does not take into account specific factors for a particular aggregation which may increase or decrease the number of equations entering a model. The limit on model size is often not the number of equations but the number of nonzeros in the matrix factorization. It is not uncommon with GTAP models to encounter applications in which there are more than 1 million nonzeros in the Jacobian and over 15 million nonzers in the associated LU factorization.<sup>19</sup>

The following sets are define in a .MAP file:

```
set    rr          Regions in aggregate model,
       ii          Sectors in aggregate model,
       ff          Factors of production in aggregate model,
       mapr(r,rr) Mapping from source to target regions
```

<sup>19</sup>It is essential to have sufficient RAM when solving large GTAP models, as solution times are measured in weeks rather than minutes when there is insufficient memory and data must be swapped to disk.

mapi(i,ii)	Mapping from source to target commodities
mapf(f,ff)	Mapping from source to target Factors of production
eec	Energy statistics in target model
mape(ec,eec)	Mapping from source to target energy stats;

Here is a sample mapping file:

```

$title Mapping from GTAP version 6

$setglobal source china

set r Regions /
  USA United States
  EUR Europe
  CHN China
  ROW Rest of world /;

set i Sectors/
  Y All goods and services /;

set f Factors of production /
  LAB Unskilled labor,
  CAP Capital /;

SET mapr(*,r) Regions /
  USA.USA United States
  EUR.EUR Europe
  CHN.CHN China
  ROW.ROW Rest of world /;

set mapi(*,i) Mapping for sectors and goods /
  AGR.Y Agriculture
  MFR.Y Manufactured goods
  SER.Y Services
  ENR.Y Energy
  CGD.Y Savings good /;

SET mapf(*,f) Factors of production /
  LND.CAP Land,
  SKL.LAB Skilled labor,
  LAB.LAB Unskilled labor,
  CAP.CAP Capital,
  RES.CAP Natural resources /

* -----
* Define energy statistics:

SET ec Energy commodities /
  col Coal
  cru Crude oil
  gas Natural gas,
  oil Refined oil products,
  ele Electricity /;

```

```

SET      mape(*,ec) /
        ecoa.col      Coal,
        eoil.cru      Crude oil,
        (egas,egdt).gas Natural gas,
        ep_c.oil      Refined oil products,
        eely.ele      Electricity  /;

```

## 5.4 Directory Contents

Here is an overview of files provided in the GTAP6inGAMS distribution:

- BUILD

**flex2gdx.gms** Stand-alone program for translating a FlexAgg dataset into gdx format. In its default operation, this program expects to find .HAR files GSDSET, GSDDAT, GSDPAR and GSDVOLE in the DATA directory. These files are combined to create GSD.GDX which is written to the DATA subdirectory.

**zip2gdx.gms** Stand-alone program for translating a GTAPAgg dataset into gdx format. In its default operation, this program expects to find CHINA.ZIP in the DATA directory. These files from this archive are extracted and then recombined to create CHINA.GDX which is written to the DATA subdirectory.

**filter.gms** This program filters small coefficients from GSD.GDX to create a sparse dataset. The syntax is

```
gams filter --tol=xxx
```

The output file is named *GSDtol.GDX* and written to the DATA directory. *tol* refers to the filter tolerance. The default filter has a value of 01, but either small or larger values could be used instead, ranging from `--tol=0.1` to `--tol=0.001`. A smaller the value of `--tol`, leaves a larger number of small numbers in the dataset, something which can lead to computational challenges in large-scale datasets.

**gtapaggr.gms** This program is a standard GAMS program which can be used to aggregate datasets. The invocation for this program is

```
gams gtapaggr --ds=xxx
```

Where *ds* indicates the name of the dataset to be created. The program looks in `%DS%.MAP` for source and mapping definitions.

**aggr.gms,checkset.gms,chktarget.gms** Utility routines used by GTAPAGGR.

**impose.gms** A GAMS program illustrating how tax rates in a GTAP6inGAMS dataset can be altered. This program calls REBALANCE.GMS.

**rebalance.gms** Performs weighted least-squares adjustment of domestic values to restore a benchmark equilibrium following alteration of base year tax rates or value shares.

**gtap6data.gms** A utility \$batinclude routine for reading a GTAP6inGAMS GDX file.

- DEFINES

Contains mapping files for aggregations of GTAP6 datasets. Files ending with .MAP define an aggregation in terms of the source dataset and mappings from sets in the source to sets in the target.

- MODELS

Contains template GAMS programs illustrating how the GTAP data can be GAMS programs. These include four model files:

**mrtmge.gms** The standard model specified as an mixed complementarity model using an MPSGE representation of demand and supply functions.

**mrtmcp.gms** The standard model specified as an mixed complementarity model using GAMS with the GAMS-F function preprocessor. The model is specified in both MCP and CNS formats.

**gtap6data.gms** The utility \$batinclude routine for reading a GTAP6inGAMS GDX file. (This file is a duplicate of the file included in the BUILD subdirectory.

**mcf.gms** A static MPSGE trade model which investigates the marginal cost of funds from different tax bases in the standard GTAP framework.


**uniform.gms** Numerical experiments with a slightly modified version of the model from MRTMGE.GMS investigate welfare and revenue impacts of uniformity of tariffs or consumption taxes. This program illustrates how to perform equal incidence (revenue-neutral) tax reform.

**fta.gms** A static MPSGE trade model with policy simulations which investigate the economic effects of free trade areas. The model used in this analysis includes a public sector budget constraint and an equal yield constraint.

**opttariff.gms** Investigates the welfare consequences of equ-proportional variations in tariff and export subsidy rates. These calculations highlight the potentially important role played by terms of trade effects in the standard GTAP model. This program call GNUPLOT to provide a graphical representation model output, as shown in 5.4

**DATA** This directory is intended to hold all data files. This distribution media contains three files, CHP13.GDX, ASA7X5.GDX and CIS.GDX. All of these are GAMSinGTAP datasets which are distributed freely with the system. CHINA.ZIP is a sample GTA-PAgg dataset which may be translated using ZIP2GDX.GMS from the BUILD directory.

Figure 9: Welfare Effects of Proportional Tariff Adjustments (% ev)



## References

- Dixit, Avinash and Victor Norman**, *International Trade Theory*, Cambridge University Press, 1992.
- Harrison, W. Jill and Ken R. Pearson**, “Computing solutions for large general equilibrium models using GEMPACK,” *Computational Economics*, 1996, 9, 83–127.
- Hertel, Thomas W., ed.**, *Global Trade Analysis: Modeling and Applications*, Cambridge University Press, 1997.
- Mathiesen, Lars**, “Computation of Economic Equilibrium by a Sequence of Linear Complementarity Problems,” *Mathematical Programming Study*, 1985, 23, 144–162.
- McDougall, Robert**, “The GTAP 6 Database,” Technical Report, Purdue University 2005.
- Rutherford, Thomas F.**, “Extensions of GAMS for Complementarity Problems,” *Journal of Economic Dynamics and Control*, 1995.
- , “GTAPinGAMS,” Working Paper, University of Colorado, Boulder 1997.
- , “MPSGE,” *Computational Economics*, 1999.
- and **Sergey Paltsev**, “GTAP-Energy in GAMS: The Dataset and Static Model,” Working Paper 00-2, University of Colorado, Boulder 2000.



Table 5: Regions

AUS	Australia	IRL	Ireland
NZL	New Zealand	ITA	Italy
XOC	Rest of Oceania	LUX	Luxembourg
CHN	China	NLD	Netherlands
HKG	Hong Kong	PRT	Portugal
JPN	Japan	ESP	Spain
KOR	Republic of Korea	SWE	Sweden
TWN	Taiwan	CHE	Switzerland
XEA	Rest of EastAsia	XEF	Rest of EFTA
IDN	Indonesia	XER	Rest of Europe
MYS	Malaysia	ALB	Albania
PHL	Philippines	BGR	Bulgaria
SGP	Singapore	HRV	Croatia
THA	Thailand	CYP	Cyprus
VNM	Viet Nam	CZE	Czech Republic
XSE	Rest of Southeast Asia	HUN	Hungary
BGD	Bangladesh	MLT	Malta
IND	India	POL	Poland
LKA	Sri Lanka	ROM	Romania
XSA	Rest of South Asia	SVK	Slovakia
CAN	Canada	SVN	Slovenia
USA	United States of America	EST	Estonia
MEX	Mexico	LVA	Latvia
XNA	Rest of North America	LTU	Lithuania
COL	Colombia	RUS	Russian Federation
PER	Peru	XSU	Rest of Former Soviet Union
VEN	Venezuela	TUR	Turkey
XAP	Rest of Andean Pact	XME	Rest of Middle East
ARG	Argentina	MAR	Morocco
BRA	Brazil	TUN	Tunisia
CHL	Chile	XNF	Rest of North Africa
URY	Uruguay	BWA	Botswana
XSM	Rest of South America	ZAF	South Africa
XCA	Central America	XSC	Rest of SACU
XFA	Rest of FTAA	MWI	Malawi
XCB	Rest of the Caribbean	MOZ	Mozambique
AUT	Austria	TZA	Tanzania
BEL	Belgium	ZMB	Zambia
DNK	Denmark	ZWE	Zimbabwe
FIN	Finland	XSD	Rest of SADC
FRA	France	MDG	Madagascar
DEU	Germany	UGA	Uganda
GBR	United Kingdom	XSS	Rest of Sub Saharan Africa
GRC	Greece		

Table 6: Commodities and Industries

PDR	Paddy rice	LUM	Wood products
WHT	Wheat	PPP	Paper products, publishing
GRO	Cereal grains nec	P_C	Petroleum, coal products
V_F	Vegetables, fruit, nuts	CRP	Chemical,rubber,plastic prods
OSD	Oil seeds	NMM	Mineral products nec
C_B	Sugar cane, sugar beet	LS	Ferrous metals
PFB	Plant-based fibers	NFM	Metals nec
OCR	Crops nec	FMP	Metal products
CTL	Cattle,sheep,goats,horses	MVH	Motor vehicles and parts
OAP	Animal products nec	OTN	Transport equipment nec
RMK	Raw milk	ELE	Electronic equipment
WOL	Wool, silk-worm cocoons	OME	Machinery and equipment nec
FRS	Forestry	OMF	Manufactures nec
FSH	Fishing	ELY	Electricity
COA	Coal	GDT	Gas manufacture, distribution
OIL	Oil	WTR	Water
GAS	Gas	CNS	Construction
OMN	Minerals nec	TRD	Trade
CMT	Meat: cattle,sheep,goats,horse	OTP	Transport nec
OMT	Meat products nec	WTP	Sea transport
VOL	Vegetable oils and fats	ATP	Air transport
MIL	Dairy products	CMN	Communication
PCR	Processed rice	OFI	Financial services nec
SGR	Sugar	ISR	Insurance
OFD	Food products nec	OBS	Business services nec
B_T	Beverages and tobacco products	ROS	Recreation and other services
TEX	Textiles	OSG	PubAdmin/Defence/Health/Educat
WAP	Wearing apparel	DWE	Dwellings
LEA	Leather products	CGD	Aggregate investment

Table 7: Primary Factors

Mobile factors:

SKL	Skilled labor
LAB	Unskilled labor
CAP	Capital

“Sluggish” factors:

LND	Land
RES	Natural resources